



*K $\omega$*  **Documentation**  
*Release 2.0.0*

June 13, 2017



<b>1</b>	<b>Overview</b>	<b>1</b>
<b>2</b>	<b>Algorithm</b>	<b>2</b>
2.1	Shifted BiCG method with seed switching technique . . . . .	2
2.2	Shifted COCG method with seed switching technique . . . . .	3
2.3	Shifted CG method with seed switching technique . . . . .	4
<b>3</b>	<b>Schematic workflow of this library</b>	<b>6</b>
3.1	The schematic workflow of shifted BiCG library . . . . .	7
3.2	The schematic workflow of shifted COCG library . . . . .	8
3.3	The schematic workflow of shifted CG library . . . . .	9
<b>4</b>	<b>Install</b>	<b>10</b>
4.1	Overall procedure . . . . .	10
4.2	Options for configure . . . . .	10
<b>5</b>	<b>Usage</b>	<b>12</b>
5.1	Details of each routines . . . . .	13
5.2	Sample codes for using shifted BiCG library . . . . .	19
<b>6</b>	<b>Re-distribution of this library</b>	<b>23</b>
6.1	Contain Komega in your program . . . . .	23
6.2	Build Komega without Autoconf . . . . .	23
6.3	Lesser General Public License . . . . .	24
<b>7</b>	<b>Contact</b>	<b>25</b>
<b>8</b>	<b>Reference</b>	<b>26</b>



---

**Overview**

---

This document is a manual for  $K\omega$  which is the library to solve the shifted linear equation within the Krylov subspace. This library provides routines to solve the following shifted linear equation (with the projection),

$$G_{ij}(z) = \langle i|(z\hat{I} - \hat{H})^{-1}|j\rangle \equiv \varphi_i^* \cdot (z\hat{I} - \hat{H})^{-1}\varphi_j. \quad (1.1)$$

The source codes of  $K\omega$  is written in FORTRAN and requires the BLAS Level 1 routines.

---

## Algorithm

---

This library provides the four kinds of numerical solvers. The kind of solvers is selected under the condition whether the Hamiltonian  $\hat{H}$  and/or the frequency  $z$  are complex or real number. It is noted that  $\hat{H}$  must be Hermitian (symmetric) for complex (real) number.

- $(\hat{H}, z) = (\text{complex}, \text{complex})$ : Shifted Bi-Conjugate Gradient(BiCG) method [1]
- $(\hat{H}, z) = (\text{real}, \text{complex})$ : Shifted Conjugate Orthogonal Conjugate Gradient(COCCG) method [2]
- $(\hat{H}, z) = (\text{complex}, \text{real})$ : Shifted Conjugate Gradient(CG) method (using complex vector)
- $(\hat{H}, z) = (\text{real}, \text{real})$ : Shifted Conjugate Gradient(CG) method (using real vector)

For above methods, seed switching [2] is adopted. Hereafter, the number of the left (right) side vector is written as  $N_L$  ( $N_R$ ). The details of each algorithm are written as follows.

### 2.1 Shifted BiCG method with seed switching technique

$$G_{ij}(z_k) = 0 (i = 1 \cdots N_L, j = 1 \cdots N_R, k = 1 \cdots N_z)$$

do  $j = 1 \cdots N_R$

$$\mathbf{r} = \boldsymbol{\varphi}_j,$$

$$\tilde{\mathbf{r}} = \text{an arbitrary vector}, \mathbf{r}^{\text{old}} = \tilde{\mathbf{r}}^{\text{old}} = \mathbf{0}$$

$$p_{ik} = 0 (i = 1 \cdots N_L, k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$$

$$\rho = \infty, \alpha = 1, z_{\text{seed}} = 0$$

do iteration

◦ Seed equation

$$\rho^{\text{old}} = \rho, \rho = \tilde{\mathbf{r}}^* \cdot \mathbf{r}$$

$$\beta = \rho / \rho^{\text{old}}$$

$$\mathbf{q} = (z_{\text{seed}} \hat{I} - \hat{H}) \mathbf{r}$$

$$\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\tilde{\mathbf{r}}^* \cdot \mathbf{q} - \beta \rho / \alpha}$$

◦ Shifted equation

do  $k = 1 \cdots N_z$

$$\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$$

do  $i = 1 \cdots N_L$

$$p_{ik} = \frac{1}{\pi_k} \varphi_i^* \cdot \mathbf{r} + \frac{\pi_k^{\text{old}} \pi_k^{\text{old}}}{\pi_k \pi_k} \beta p_{ik}$$

$$G_{ij}(z_k) = G_{ij}(z_k) + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha p_{ik}$$

$$\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$$

end do  $i$

end do  $k$

$$\mathbf{q} = \left(1 + \frac{\alpha\beta}{\alpha^{\text{old}}}\right) \mathbf{r} - \alpha \mathbf{q} - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{r}^{\text{old}}, \mathbf{r}^{\text{old}} = \mathbf{r}, \mathbf{r} = \mathbf{q}$$

$$\mathbf{q} = (z_{\text{seed}}^* \hat{I} - \hat{H}) \tilde{\mathbf{r}}, \mathbf{q} = \left(1 + \frac{\alpha^* \beta^*}{\alpha^{\text{old}*}}\right) \tilde{\mathbf{r}} - \alpha^* \mathbf{q} - \frac{\alpha^* \beta^*}{\alpha^{\text{old}*}} \tilde{\mathbf{r}}^{\text{old}}, \tilde{\mathbf{r}}^{\text{old}} = \tilde{\mathbf{r}}, \tilde{\mathbf{r}} = \mathbf{q}$$

o Seed switch

Search  $k$  which gives the smallest  $|\pi_k|$ .  $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$

$$\mathbf{r} = \mathbf{r}/\pi_{\text{seed}}, \mathbf{r}^{\text{old}} = \mathbf{r}^{\text{old}}/\pi_{\text{seed}}^{\text{old}}, \tilde{\mathbf{r}} = \tilde{\mathbf{r}}/\pi_{\text{seed}}^*, \tilde{\mathbf{r}}^{\text{old}} = \tilde{\mathbf{r}}^{\text{old}}/\pi_{\text{seed}}^{\text{old}*}$$

$$\alpha = (\pi_{\text{seed}}^{\text{old}}/\pi_{\text{seed}})\alpha, \rho = \rho/(\pi_{\text{seed}}^{\text{old}}\pi_{\text{seed}})$$

$$\{\pi_k = \pi_k/\pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}}/\pi_{\text{seed}}^{\text{old}}\}$$

if(  $|\mathbf{r}| < \text{Threshold}$ ) exit

end do iteration

end do  $j$

## 2.2 Shifted COCG method with seed switching technique

This method is obtained by  $\tilde{\mathbf{r}} = \mathbf{r}^*$ ,  $\tilde{\mathbf{r}}^{\text{old}} = \mathbf{r}^{\text{old}*}$  in the BiCG method.

$$G_{ij}(z_k) = 0 (i = 1 \cdots N_L, j = 1 \cdots N_R, k = 1 \cdots N_z)$$

do  $j = 1 \cdots N_R$

$$\mathbf{r} = \varphi_j, \mathbf{r}^{\text{old}} = \mathbf{0}$$

$$p_{ik} = 0 (i = 1 \cdots N_L, k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$$

$$\rho = \infty, \alpha = 1, z_{\text{seed}} = 0$$

do iteration

o Seed equation

$$\rho^{\text{old}} = \rho, \rho = \mathbf{r} \cdot \mathbf{r}$$

$$\beta = \rho/\rho^{\text{old}}$$

$$\mathbf{q} = (z_{\text{seed}} \hat{I} - \hat{H}) \mathbf{r}$$

$$\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\mathbf{r} \cdot \mathbf{q} - \beta \rho / \alpha}$$

o Shifted equation

do  $k = 1 \cdots N_z$

```

 $\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$ 
do  $i = 1 \cdots N_L$ 
   $p_{ik} = \frac{1}{\pi_k} \varphi_i^* \cdot \mathbf{r} + \frac{\pi_k^{\text{old}} \pi_k}{\pi_k \pi_k} \beta p_{ik}$ 
   $G_{ij}(z_k) = G_{ij}(z_k) + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha p_{ik}$ 
   $\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$ 
end do  $i$ 
end do  $k$ 
 $\mathbf{q} = \left(1 + \frac{\alpha\beta}{\alpha^{\text{old}}}\right) \mathbf{r} - \alpha \mathbf{q} - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{r}^{\text{old}}, \mathbf{r}^{\text{old}} = \mathbf{r}, \mathbf{r} = \mathbf{q}$ 
o Seed switch
Search  $k$  which gives the smallest  $|\pi_k|$ .  $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$ 
 $\mathbf{r} = \mathbf{r}/\pi_{\text{seed}}, \mathbf{r}^{\text{old}} = \mathbf{r}^{\text{old}}/\pi_{\text{seed}}^{\text{old}}$ 
 $\alpha = (\pi_{\text{seed}}^{\text{old}}/\pi_{\text{seed}})\alpha, \rho = \rho/(\pi_{\text{seed}}^{\text{old}}\pi_{\text{seed}}^{\text{old}})$ 
 $\{\pi_k = \pi_k/\pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}}/\pi_{\text{seed}}^{\text{old}}\}$ 
if(  $|\mathbf{r}| < \text{Threshold}$ ) exit
end do iteration
end do  $j$ 

```

## 2.3 Shifted CG method with seed switching technique

This method is obtained by  $\tilde{\mathbf{r}} = \mathbf{r}, \tilde{\mathbf{r}}^{\text{old}} = \mathbf{r}^{\text{old}}$  in the BiCG method.

$G_{ij}(z_k) = 0 (i = 1 \cdots N_L, j = 1 \cdots N_R, k = 1 \cdots N_z)$

do  $j = 1 \cdots N_R$

$\mathbf{r} = \varphi_j, \mathbf{r}^{\text{old}} = \mathbf{0}$

$p_{ik} = 0 (i = 1 \cdots N_L, k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$

$\rho = \infty, \alpha = 1, z_{\text{seed}} = 0$

do iteration

o Seed equation

$\rho^{\text{old}} = \rho, \rho = \mathbf{r}^* \cdot \mathbf{r}$

$\beta = \rho/\rho^{\text{old}}$

$\mathbf{q} = (z_{\text{seed}}\hat{I} - \hat{H})\mathbf{r}$

$\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\mathbf{r}^* \cdot \mathbf{q} - \beta\rho/\alpha}$

o Shifted equation

do  $k = 1 \cdots N_z$

$\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$

do  $i = 1 \cdots N_L$



$$p_{ik} = \frac{1}{\pi_k} \varphi_i^* \cdot \mathbf{r} + \left( \frac{\pi_k^{\text{old}}}{\pi_k} \right)^2 \beta p_{ik}$$

$$G_{ij}(z_k) = G_{ij}(z_k) + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha p_{ik}$$

$$\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$$

end do  $i$

end do  $k$

$$\mathbf{q} = \left( 1 + \frac{\alpha\beta}{\alpha^{\text{old}}} \right) \mathbf{r} - \alpha \mathbf{q} - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{r}^{\text{old}}, \mathbf{r}^{\text{old}} = \mathbf{r}, \mathbf{r} = \mathbf{q}$$

o Seed switch

Search  $k$  which gives the minimum value of  $|\pi_k|$ .  $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$

$$\mathbf{r} = \mathbf{r} / \pi_{\text{seed}}, \mathbf{r}^{\text{old}} = \mathbf{r}^{\text{old}} / \pi_{\text{seed}}^{\text{old}}$$

$$\alpha = (\pi_{\text{seed}}^{\text{old}} / \pi_{\text{seed}}) \alpha, \rho = \rho / \pi_{\text{seed}}^{\text{old}^2}$$

$$\{ \pi_k = \pi_k / \pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}} / \pi_{\text{seed}}^{\text{old}} \}$$

if(  $|\mathbf{r}| < \text{Threshold}$ ) exit

end do iteration

end do  $j$

---

## Schematic workflow of this library

---

In the following description, the loop for  $N_R$  is omitted for simplicity and instead of  $G_{ij}(z_k)$ , the  $N_L$ -dimensional vector  $\mathbf{x}_k$  is obtained by using the library.

The names of the routines is defined as follows.

- `komega_bicg_init`, `komega_cocg_init`, `komega_cg_c_init`, `komega_cg_r_init`  
Set the initial conditions such as the allocation of variables used in the library.
- `komega_bicg_update`, `komega_cocg_update`, `komega_cg_c_update`,  
`komega_cg_r_update`  
These routines are called in the iteration to update the solution vectors.
- `komega_bicg_finalize`, `komega_cocg_finalize`, `komega_cg_c_finalize`,  
`komega_cg_r_finalize`  
Release the allocated vectors in the library.
- `komega_bicg_getcoef`, `komega_cocg_getcoef`, `komega_cg_c_getcoef`,  
`komega_cg_r_getcoef`  
Get the  $\alpha$ ,  $\beta$ ,  $z_{\text{seed}}$ ,  $\mathbf{r}^L$  conserved at each iteration.
- `komega_bicg_getvec`, `komega_cocg_getvec`, `komega_cg_c_getvec`,  
`komega_cg_r_getvec`  
Get the vectors  $\mathbf{r}$ ,  $\mathbf{r}^{\text{old}}$ ,  $\tilde{\mathbf{r}}$ ,  $\tilde{\mathbf{r}}^{\text{old}}$ .
- `komega_bicg_restart`, `komega_cocg_restart`, `komega_cg_c_restart`, `CG_R_restart`

---

**Note:**

- Give the vector size  $N_H$  corresponding to the size of the Hilbert space and the number of the frequency  $z$ .
- Allocate the two vectors (in the case of BiCG method, four vectors) with the size of  $N_H$ .
- Give the function for the Hamiltonian-vector production.
- Allocate the solution vectors. It is noted that the length of each solution vector is not always equal to  $N_H$ . In fact, the its length in the previous section is  $N_L$ . In this case, the length of the (bi-)conjugate gradient vector  $\mathbf{p}_k$  ( $k = 1, \dots, N_z$ ) also becomes  $N_L$ . We have to prepare a code for projecting  $N_H$ -dimensional vector onto  $N_L$  dimensional space.

$$\mathbf{r}^L = \hat{P}^\dagger \mathbf{r}, \quad \hat{P} \equiv (\varphi_1, \dots, \varphi_{N_L})$$

- If the result converges (or a breakdown occurs), `komega_*_update` return the first element of `status` as a negative integer. Therefore, please exit loop when `status(1) < 0`.

- The 2-norm is used for the convergence check in the routine `komega*_update`. Therefore, if 2-norms of residual vectors at all shift points becomes smaller than `threshold`, this routine assumes the result is converged.
- We can obtain the history of  $\alpha, \beta, \mathbf{r}^L$  for restarting calculation. In this case, `itermax` must not be 0.

### 3.1 The schematic workflow of shifted BiCG library

```

Allocate  $\mathbf{v}_{12}, \mathbf{v}_{13}, \mathbf{v}_2, \mathbf{v}_3, \{\mathbf{x}_k\}, \mathbf{r}^L \mathbf{v}_2 = \varphi_j$ 
komega_bicg_init(N_H, N_L, N_z, x, z, itermax, threshold) start
    Allocate  $\mathbf{v}_3, \mathbf{v}_5, \{\pi_k\}, \{\pi_k^{\text{old}}\}, \{\mathbf{p}_k\}$ 
    Copy  $\{z_k\}$ 
    If itermax  $\neq 0$ , allocate arrays to store  $\alpha, \beta$ , and  $\mathbf{r}^L$  at each iteration.
     $\mathbf{v}_4 = \mathbf{v}_2^*$  (an arbitrary vector),  $\mathbf{v}_3 = \mathbf{v}_5 = \mathbf{0}$ ,
     $\mathbf{p}_k = \mathbf{x}_k = \mathbf{0} (k = 1 \cdots N_z)$ ,  $\pi_k = \pi_k^{\text{old}} = 1 (k = 1 \cdots N_z)$ 
     $\rho = \infty, \alpha = 1, z_{\text{seed}} = 0$ 
    ( $\mathbf{v}_2 \equiv \mathbf{r}, \mathbf{v}_3 \equiv \mathbf{r}^{\text{old}}, \mathbf{v}_4 \equiv \tilde{\mathbf{r}}, \mathbf{v}_5 \equiv \tilde{\mathbf{r}}^{\text{old}}$ .)
komega_bicg_init finish
do iteration
     $\mathbf{r}^L = \hat{P}^\dagger \mathbf{v}_2$ 
     $\mathbf{v}_{12} = \hat{H} \mathbf{v}_2, \mathbf{v}_{14} = \hat{H} \mathbf{v}_4$  [Or  $(\mathbf{v}_{12}, \mathbf{v}_{14}) = \hat{H}(\mathbf{v}_2, \mathbf{v}_4)$ ]
    komega_bicg_update(v_12, v_2, v_14, v_4, x, r_small, status) start
        ◦ Seed equation
         $\rho^{\text{old}} = \rho, \rho = \mathbf{v}_4^* \cdot \mathbf{v}_2$ 
         $\beta = \rho / \rho^{\text{old}}$ 
         $\mathbf{v}_{12} = z_{\text{seed}} \mathbf{v}_2 - \mathbf{v}_{12}, \mathbf{v}_{14} = z_{\text{seed}}^* \mathbf{v}_4 - \mathbf{v}_{14}$ 
         $\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\mathbf{v}_3^* \cdot \mathbf{v}_{12} - \beta \rho / \alpha}$ 
        ◦ Shifted equation
        do  $k = 1 \cdots N_z$ 
             $\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})] \pi_k - \frac{\alpha \beta}{\alpha^{\text{old}}} (\pi_k^{\text{old}} - \pi_k)$ 
             $\mathbf{p}_k = \frac{1}{\pi_k} \mathbf{r}^L + \frac{\pi_k^{\text{old}} \pi_k^{\text{old}}}{\pi_k \pi_k} \beta \mathbf{p}_k$ 
             $\mathbf{x}_k = \mathbf{x}_k + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha \mathbf{p}_k$ 
             $\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$ 
        end do  $k$ 
         $\mathbf{v}_{12} = \left(1 + \frac{\alpha \beta}{\alpha^{\text{old}}}\right) \mathbf{v}_2 - \alpha \mathbf{v}_{12} - \frac{\alpha \beta}{\alpha^{\text{old}}} \mathbf{v}_3, \mathbf{v}_3 = \mathbf{v}_2, \mathbf{v}_2 = \mathbf{v}_{12}$ 
         $\mathbf{v}_{14} = \left(1 + \frac{\alpha^* \beta^*}{\alpha^{\text{old}*}}\right) \mathbf{v}_4 - \alpha^* \mathbf{v}_{14} - \frac{\alpha^* \beta^*}{\alpha^{\text{old}*}} \mathbf{v}_5, \mathbf{v}_5 = \mathbf{v}_4, \mathbf{v}_4 = \mathbf{v}_{14}$ 

```

```

    ◦ Seed switch
    Search  $k$  which gives the smallest  $|\pi_k| \rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$ 
     $\mathbf{v}_2 = \mathbf{v}_2/\pi_{\text{seed}}, \mathbf{v}_3 = \mathbf{v}_3/\pi_{\text{seed}}^{\text{old}}, \mathbf{v}_4 = \mathbf{v}_4/\pi_{\text{seed}}^*, \mathbf{v}_5 = \mathbf{v}_5/\pi_{\text{seed}}^{\text{old}*}$ 
     $\alpha = (\pi_{\text{seed}}^{\text{old}}/\pi_{\text{seed}})\alpha, \rho = \rho/(\pi_{\text{seed}}^{\text{old}}\pi_{\text{seed}}^{\text{old}})$ 
     $\{\pi_k = \pi_k/\pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}}/\pi_{\text{seed}}^{\text{old}}\}$ 
komega_bicg_update finish
if(status(1) < 0 (This indicates  $|\mathbf{v}_2| < \text{Threshold}$ )) exit
end do iteration
komega_bicg_finalize start
    Deallocate  $\mathbf{v}_4, \mathbf{v}_5, \{\pi_k\}, \{\pi_k^{\text{old}}\}, \{\mathbf{p}_k\}$ 
komega_bicg_finalize finish

```

## 3.2 The schematic workflow of shifted COCG library

```

Allocate  $\mathbf{v}_1, \mathbf{v}_2, \{\mathbf{x}_k\}, \mathbf{r}^L, \mathbf{v}_2 = \boldsymbol{\varphi}_j$ 
komega_cocg_init(N_H, N_L, N_z, x, z, itermax, threshold) start
    Allocate  $\mathbf{v}_3, \{\pi_k\}, \{\pi_k^{\text{old}}\}, \{\mathbf{p}_k\}$ 
    Copy  $\{z_k\}$ 
    If itermax  $\neq 0$ , allocate arrays to store  $\alpha, \beta$ , and  $\mathbf{r}^L$ .
     $\mathbf{v}_3 = \mathbf{0}$ ,
     $\mathbf{p}_k = \mathbf{x}_k = \mathbf{0}(k = 1 \cdots N_z), \pi_k = \pi_k^{\text{old}} = 1(k = 1 \cdots N_z)$ 
     $\rho = \infty, \alpha = 1, \beta = 0, z_{\text{seed}} = 0$ 
    ( $\mathbf{v}_2 \equiv \mathbf{r}, \mathbf{v}_3 \equiv \mathbf{r}^{\text{old}}$ .)
komega_cocg_init finish
do iteration
     $\mathbf{r}^L = \hat{P}^\dagger \mathbf{v}_2$ 
     $\mathbf{v}_1 = \hat{H} \mathbf{v}_2$ 
    komega_cocg_update(v_1, v_2, x, r_small, status) start
        ◦ Seed equationw
         $\rho^{\text{old}} = \rho, \rho = \mathbf{v}_2 \cdot \mathbf{v}_2$ 
         $\beta = \rho/\rho^{\text{old}}$ 
         $\mathbf{v}_1 = z_{\text{seed}}\mathbf{v}_2 - \mathbf{v}_1$ 
         $\alpha^{\text{old}} = \alpha, \alpha = \frac{\rho}{\mathbf{v}_2 \cdot \mathbf{v}_1 - \beta\rho/\alpha}$ 
        ◦ Shifted equations
        do  $k = 1 \cdots N_z$ 

```

---


$$\pi_k^{\text{new}} = [1 + \alpha(z_k - z_{\text{seed}})]\pi_k - \frac{\alpha\beta}{\alpha^{\text{old}}}(\pi_k^{\text{old}} - \pi_k)$$

$$\mathbf{p}_k = \frac{1}{\pi_k} \mathbf{r}^L + \frac{\pi_k^{\text{old}} \pi_k^{\text{old}}}{\pi_k \pi_k} \beta \mathbf{p}_k$$

$$\mathbf{x}_k = \mathbf{x}_k + \frac{\pi_k}{\pi_k^{\text{new}}} \alpha \mathbf{p}_k$$

$$\pi_k^{\text{old}} = \pi_k, \pi_k = \pi_k^{\text{new}}$$

end do  $k$

$$\mathbf{v}_1 = \left(1 + \frac{\alpha\beta}{\alpha^{\text{old}}}\right) \mathbf{v}_2 - \alpha \mathbf{v}_1 - \frac{\alpha\beta}{\alpha^{\text{old}}} \mathbf{v}_3$$

$$\mathbf{v}_3 = \mathbf{v}_2, \mathbf{v}_2 = \mathbf{v}_1$$

o Seed switch

Search  $k$  which gives the smallest  $|p_{i_k}|$ .  $\rightarrow z_{\text{seed}}, \pi_{\text{seed}}, \pi_{\text{seed}}^{\text{old}}$

$$\mathbf{v}_2 = \mathbf{v}_2 / \pi_{\text{seed}}, \mathbf{v}_3 = \mathbf{v}_3 / \pi_{\text{seed}}^{\text{old}}$$

$$\alpha = (\pi_{\text{seed}}^{\text{old}} / \pi_{\text{seed}}) \alpha, \rho = \rho / (\pi_{\text{seed}}^{\text{old}} \pi_{\text{seed}}^{\text{old}})$$

$$\{\pi_k = \pi_k / \pi_{\text{seed}}, \pi_k^{\text{old}} = \pi_k^{\text{old}} / \pi_{\text{seed}}^{\text{old}}\}$$

komega\_cocg\_update finish

if(status(1) < 0 (This indicates  $|\mathbf{v}_2| < \text{Threshold.}$ )) exit

end do iteration

komega\_cocg\_finalize start

Deallocate  $\mathbf{v}_3, \{\pi_k\}, \{\pi_k^{\text{old}}\}, \{\mathbf{p}_k\}$

komega\_cocg\_finalize finish

### 3.3 The schematic workflow of shifted CG library

The workflow is the same as that of the shifted COCG library.

## 4.1 Overall procedure

First, please type

```
$ ./configure --prefix=install_dir
```

Then, this script checks the compiler and the libraries required for the installation, and creates Makefiles. `install_dir` indicates the full path of the directory where the library is installed (you should replace it according to your case). If none is specified, `/usr/local/` is chosen for storing libraries by `make install` (Therefore, if one is not the admin, `install_dir` must be specified to the different directory). `configure` has many options, and they are used according to the environment etc. For more details, please see *Options for configure*.

After `configure` finishes successfully and Makefiles are generated, please type

```
$ make
```

to build libraries. Then please type

```
$ make install
```

to store libraries and the sample program to `install_dir/lib` and `install_dir/bin`, respectively. Although one can use libraries and the sample program without `make install`, they are a little different to the installed one.

Add the *K $\omega$*  library directory (`install_dir/lib`) to the search path of the dynamically linked program (environment variable `LD_LIBRARY_PATH`).

```
$ export LD_LIBRARY_PATH=${LD_LIBRARY_PATH}:install_dir/lib
```

## 4.2 Options for configure

`configure` has many options and environment variables. They can be specified at once. E.g.

```
$ ./configure --prefix=/home/komega/ --with-mpi=yes FC=mpif90
```

All options and variables have default values. We show a part of them as follows:

```
---prefix
```

Default: `---prefix=/usr/local/`. Specify the directory where the library etc. are installed.

```
--with-mpi
```

Default: `--with-mpi=no` (without MPI). Whether use MPI (`--with-mpi=yes`), or not.

`--with-openmp`

Default: `--with-openmp=yes` (with OpenMP). Whether use OpenMP or not (`--with-openmp=no`).

`--enable-shared`

Default: `--enable-shared`. Whether generate shared library.

`--enable-static`

Default: `--enable-static`. Whether generate static library.

`--disable-zdot`

Default: `--enable-zdot`. When ZDOTC and ZDOTU in BLAS do not work correctly (e.g. standard BLAS in MacOSX), please use this option to be disable these functions.

`--enable-threadsaf`

Default: `--disable-threadsaf`. If you want to call K $\omega$  routine in the parallel region (i.e. plan to solve different equations among threads), please use this option (**Experimental**).

FC

Default: The fortran compiler chosen automatically from those in the system. When `--with-mpi` is specified, the corresponding MPI compiler (such as `mpif90`) is searched. If FC printed the end of the standard-output of `configure` is not what you want, please set it manually as `./configure FC=gfortran`.

`--help`

Display all options including above, and stop without configuration.

---

## Usage

---

The calculation is done to utilize functions by the following procedures.

- Initialization (*\*\_init*)
- Update results iteratively (*\*\_update*)
- (Optional) Take the information for the restart (*\*\_getcoef*, *\*\_getvec*)
- Finalization (*\*\_finalize*)

The restart calculation can be done by the following procedures.

- Initialization with the information of the previous calculation (*\*\_restart*)
- Update results iteratively (*\*\_update*)
- (Optional) Take the information for the further restart (*\*\_getcoef*, *\*\_getvec*)
- Finalization (*\*\_finalize*)

**Warning:** Since  $K\omega$  is **not** thread safe, these routine must be called from the outside of the OpenMP-parallel region. If you want to call  $K\omega$  routine in the parallel region (i.e. plan to solve different equations among threads), please use `--enable-threadsaf` option of configure (**Experimental**).

For FORTRAN, the modules can be called by

```
USE komega_cg_r ! Conjugate-gradient method for real vectors
USE komega_cg_c ! Conjugate-gradient method for complex vectors
USE komega_cocg ! Conjugate-orthogonal conjugate-gradient mehod
USE komega_bicg ! Biconjugate-gradient method
```

When we call  $K\omega$  from C/C++ codes, we should include the header file as

```
#include komega.h
```

Scalar arguments should be passed as pointers. For MPI/Hybrid parallelized routine, the above line becomes Also, the communicator argument for the routine should be transformed from the C/C++'s one to the fortran's one as follows.

```
comm_f = MPI_Comm_c2f(comm_c);
```



## 5.1 Details of each routines

### 5.1.1 \*\_init

Set and initialize internal variables in libraries. These routines should be called first before solving the shifted equation.

Syntax

Fortran

```
CALL komega_cg_r_init(ndim, nl, nz, x, z, itermax, threshold, comm)
CALL komega_cg_c_init(ndim, nl, nz, x, z, itermax, threshold, comm)
CALL komega_cocg_init(ndim, nl, nz, x, z, itermax, threshold, comm)
CALL komega_bicg_init(ndim, nl, nz, x, z, itermax, threshold, comm)
```

C/C++

```
komega_cg_r_init(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm);
komega_cg_c_init(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm);
komega_cocg_init(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm);
komega_bicg_init(&ndim, &nl, &nz, x, z, &itermax, &threshold, &comm);
```

Parameters

**INTEGER, INTENT (IN) :: ndim**

The dimension of solution vectors for the linearized equation. ndim for the dimension of variables in other routine is equal to this.

**INTEGER, INTENT (IN) :: nl**

The dimension of projected solution vectors. nl for the dimension of variables in other routine is equal to this.

**INTEGER, INTENT (IN) :: nz**

The number of shifted points. nz for the dimension of variables in other routine is equal to this.

**REAL (8), INTENT (OUT) :: x(nl\*nz) ! (for "CG\_R\_init", "CG\_C\_init")**

**COMPLEX (8), INTENT (OUT) :: x(nl\*nz) ! (for other cases)**

The solution vector. In this procedure, 0 vector is returned.

**REAL (8), INTENT (IN) :: z(nz) ! (for "CG\_R\_init", "CG\_C\_init")**

**COMPLEX (8), INTENT (IN) :: z(nz) ! (for other cases)**

Shifted points.

**INTEGER, INTENT (IN) :: itermax**

The maximum iteration number for allocating arrays for the restart calculation. When itermax=0, these arrays are not allocated, and the restart calculation described later becomes unavailable.

**REAL (8), INTENT (IN) :: threshold**

The threshold value for the convergence determination. When the 2-norm of the residual vector for the seed equation becomes smaller than this value, the calculation is finished.

**INTEGER, INTENT (IN), OPTIONAL** :: comm

Optional argument. Communicators for MPI such as MPI\_COMM\_WORLD. Only for MPI / Hybrid parallelization. For C compiler, just pass NULL to omit this argument.

### 5.1.2 \*\_restart

For the restart calculation, these routines are used instead of \*\_init. Set and initialize internal variables in libraries. These routines should be called first before solving the shifted equation.

Syntax

Fortran

```
CALL komega_cg_r_restart(ndim, nl, nz, x, z, itermax, threshold, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save, comm)
CALL komega_cg_c_restart(ndim, nl, nz, x, z, itermax, threshold, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save, comm)
CALL komega_cocg_restart(ndim, nl, nz, x, z, itermax, threshold, status, &
& iter_old, v2, v12, alpha_save, beta_save, z_seed, r_l_save, comm)
CALL komega_bicg_restart(ndim, nl, nz, x, z, itermax, threshold, status, &
& iter_old, v2, v12, v4, v14, alpha_save, beta_save, &
& z_seed, r_l_save, comm)
```

C/C++

```
komega_cg_r_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, status, &
& &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save, &comm);
komega_cg_c_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, status, &
& &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save, &comm);
komega_cocg_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, status, &
& &iter_old, v2, v12, alpha_save, beta_save, &z_seed, r_l_save, &comm);
komega_bicg_restart(&ndim, &nl, &nz, x, z, &itermax, &threshold, status, &
& &iter_old, v2, v12, v4, v14, alpha_save, beta_save, &
& &z_seed, r_l_save, &comm);
```

Parameters

**INTEGER, INTENT (IN)** :: ndim  
**INTEGER, INTENT (IN)** :: nl  
**INTEGER, INTENT (IN)** :: nz  
**REAL (8), INTENT (OUT)** :: x(nl\*nz)  
**REAL (8), INTENT (IN)** :: z(nz) ! (for "CG\_R\_restart", "CG\_C\_restart")  
**COMPLEX (8), INTENT (IN)** :: z(nz) ! (Other)  
**INTEGER, INTENT (IN)** :: itermax  
**REAL (8), INTENT (IN)** :: threshold  
**INTEGER, INTENT (IN), OPTIONAL** :: comm

The definition is same as \*\_init. See the parameters in \*\_init.

**INTEGER, INTENT (OUT)** :: status(3)

The error code is returned.

First component(status(1))

If the solution is converged or a breakdown occurs, the current total number of iteration with the minus sign is returned. In other cases, this routine returns the current total number of iteration. The calculation is continuable only when status(1)

is the positive value; otherwise the result is meaningless even if the calculation is continued.

Second component(status(2))

1 is returned if `itermax` is set as a finite value and the convergence condition is not satisfied at the `itermax`-th iteration. 2 is returned if  $\alpha$  diverges. 3 is returned if  $\pi_{\text{seed}}$  becomes 0. In the case of `COCG_restart` or `BiCG_restart`, 4 is returned if the residual vector and the shadow residual vector are orthogonal. In other cases, 0 is returned.

Third component(status(3))

The index of the seed point is returned.

**INTEGER**, **INTENT**(IN) :: `iter_old`

The number of iteration for the previous calculation.

**REAL**(8), **INTENT**(IN) :: `v2(ndim)` ! (for "CG\_R\_restart")  
**COMPLEX**(8), **INTENT**(IN) :: `v2(ndim)` ! (Other)

The residual vector at the last step for the previous calculation.

**REAL**(8), **INTENT**(IN) :: `v12(ndim)` ! (for "CG\_R\_restart")  
**COMPLEX**(8), **INTENT**(IN) :: `v12(ndim)` ! (Other)

The residual vector at the second from the last step for the previous calculation.

**REAL**(8), **INTENT**(IN) :: `alpha_save(iter_old)` ! (for "CG\_R\_restart", "CG\_C\_restart")  
**COMPLEX**(8), **INTENT**(IN) :: `alpha_save(iter_old)` ! (Other)

The parameters  $\alpha$  obtained by the previous calculation at each steps by (Bi)CG methods.

**REAL**(8), **INTENT**(IN) :: `beta_save(iter_old)` ! (for "CG\_R\_restart", "CG\_C\_restart")  
**COMPLEX**(8), **INTENT**(IN) :: `beta_save(iter_old)` ! (Other)

The parameters  $\beta$  obtained by the previous calculation at each steps by (Bi)CG methods.

**REAL**(8), **INTENT**(IN) :: `z_seed` ! (for "CG\_R\_restart", "CG\_C\_restart")  
**COMPLEX**(8), **INTENT**(IN) :: `z_seed` ! (Other)

The value of the seed shift for the previous calculation.

**REAL**(8), **INTENT**(IN) :: `r_l_save(nl, iter_old)` ! (for "CG\_R\_restart")  
**COMPLEX**(8), **INTENT**(IN) :: `r_l_save(nl, iter_old)` ! (Other)

The projected residual vector at each iteration for the previous calculation.

**REAL**(8), **INTENT**(IN) :: `v4(ndim)` ! (for "CG\_R\_restart")  
**COMPLEX**(8), **INTENT**(IN) :: `v4(ndim)` ! (Other)

The shadow residual vector at the last step for the previous calculation.

**REAL**(8), **INTENT**(IN) :: `v14(ndim)` ! (for "CG\_R\_restart")  
**COMPLEX**(8), **INTENT**(IN) :: `v14(ndim)` ! (Other)

The shadow residual vector at the second last step for the previous calculation.

### 5.1.3 \*\_update

It is called alternately with the matrix-vector product in the loop and updates the solution.

## Syntax

## Fortran

```
CALL komega_cg_r_update(v12, v2, x, r_l, status)
CALL komega_cg_c_update(v12, v2, x, r_l, status)
CALL komega_cocg_update(v12, v2, x, r_l, status)
CALL komega_bicg_update(v12, v2, v14, v4, x, r_l, status)
```

## C/C++

```
komega_cg_r_update(v12, v2, x, r_l, status);
komega_cg_c_update(v12, v2, x, r_l, status);
komega_cocg_update(v12, v2, x, r_l, status);
komega_bicg_update(v12, v2, v14, v4, x, r_l, status);
```

## Parameters

```
REAL(8), INTENT(INOUT) :: v12(ndim) ! (for "CG_R_update")
COMPLEX(8), INTENT(INOUT) :: v12(ndim) ! (Other)
```

The product of the residual vector ( $v2$ ) and the matrix. This routine returns the 2-norm of the updated residual vector as a first element of this array. This returned value is used, for examples, for printing the convergence profile.

```
REAL(8), INTENT(INOUT) :: v2(ndim) ! (for "CG_R_update")
COMPLEX(8), INTENT(INOUT) :: v2(ndim) ! (Other)
```

The residual vector is input and the updated residual vector is output.

```
REAL(8), INTENT(IN) :: v14(ndim) ! (for "CG_R_update")
COMPLEX(8), INTENT(IN) :: v14(ndim) ! (Other)
```

The product of the shadow residual vector ( $v4$ ) and the matrix is input.

```
REAL(8), INTENT(INOUT) :: v4(ndim) ! (for "CG_R_update")
COMPLEX(8), INTENT(INOUT) :: v4(ndim) ! (Other)
```

The shadow residual vector is input and the updated vector is output.

```
INTEGER, INTENT(OUT) :: status(3)
```

The error code is returned.

First component ( $status(1)$ )

If the solution is converged or a breakdown occurs, the current total number of iteration with the minus sign is returned. In other cases, this routine returns the current total number of iteration. The calculation is continuable only when  $status(1)$  is the positive value; otherwise the result is meaningless even if the calculation is continued.

Second component ( $status(2)$ )

1 is returned if  $itermax$  is set as a finite value in the *\*\_init* routine and the convergence condition is not satisfied at the  $itermax$ -th iteration. 2 is returned if  $\alpha$  diverges. 3 is returned if  $\pi_{seed}$  becomes 0. In the case of *COCG\_update* or *BiCG\_update*, 4 is returned if the residual vector and the shadow residual vector are orthogonal. In other cases, 0 is returned.

Third component ( $status(3)$ )

The index of the seed point is returned.

### 5.1.4 \*\_getcoef

Get the coefficients used in the restart calculation. To call these routines, `itermax` in `*_init` routine must not be 0 .

The total number of iteration (`iter_old`) used in this routine is computed by using `status` which is an output of `*_update` as follows:

```
iter_old = ABS(status(1))
```

Syntax

Fortran

```
CALL komega_cg_r_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL komega_cg_c_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL komega_cocg_getcoef(alpha_save, beta_save, z_seed, r_l_save)
CALL komega_bicg_getcoef(alpha_save, beta_save, z_seed, r_l_save)
```

C/C++

```
komega_cg_r_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
komega_cg_c_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
komega_cocg_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
komega_bicg_getcoef(alpha_save, beta_save, &z_seed, r_l_save);
```

Parameters

```
REAL(8), INTENT(OUT) :: alpha_save(iter_old) ! (for "CG_R_restart", "CG_C_restart")
COMPLEX(8), INTENT(OUT) :: alpha_save(iter_old) ! (Other)
```

The parameters  $\alpha$  of the (Bi)CG method at each iteration.

```
REAL(8), INTENT(OUT) :: beta_save(iter_old) ! (for "CG_R_restart", "CG_C_restart")
COMPLEX(8), INTENT(OUT) :: beta_save(iter_old) ! (Other)
```

The parameters  $\beta$  of the (Bi)CG method at each iteration.

```
REAL(8), INTENT(OUT) :: z_seed ! (for "CG_R_restart", "CG_C_restart")
COMPLEX(8), INTENT(OUT) :: z_seed ! (Other)
```

Seed shift.

```
REAL(8), INTENT(IN) :: r_l_save(nl, iter_old) ! ("CG_R_restart")
COMPLEX(8), INTENT(IN) :: r_l_save(nl, iter_old) ! (Other)
```

The projected residual vectors at each iteration.

### 5.1.5 \*\_getvec

Get the residual vectors to use the restart calculation. To call these routines, `itermax` in the `*_init` routine must not be 0.

Syntax

Fortran

```
CALL komega_cg_r_getvec(r_old)
CALL komega_cg_c_getvec(r_old)
CALL komega_cocg_getvec(r_old)
CALL komega_bicg_getvec(r_old, r_tilde_old)
```

C/C++

```
komega_cg_r_getvec(r_old);  
komega_cg_c_getvec(r_old);  
komega_cocg_getvec(r_old);  
komega_bicg_getvec(r_old, r_tilde_old);
```

Parameters

```
REAL(8), INTENT(OUT) :: r_old(ndim) ! (for "CG_R_getvec")  
COMPLEX(8), INTENT(OUT) :: r_tilde_old(ndim) ! (Other)
```

The residual vector at the second last step in the previous calculation.

```
COMPLEX(8), INTENT(OUT) :: r_tilde_old(ndim)
```

The shadow residual vector at the second last step in the previous calculation.

### 5.1.6 \*\_getresidual

Get the values of 2-norm of the residual vector at each shift points. These routines can be called from anywhere between \*\_init and \*\_finalize. These routines do not affect the calculation results.

Syntax

Fortran

```
CALL komega_cg_r_getresidual(res)  
CALL komega_cg_c_getresidual(res)  
CALL komega_cocg_getresidual(res)  
CALL komega_bicg_getresidual(res)
```

C/C++

```
komega_cg_r_getresidual(res);  
komega_cg_c_getresidual(res);  
komega_cocg_getresidual(res);  
komega_bicg_getresidual(res);
```

Parameters

```
COMPLEX(8), INTENT(OUT) :: res(nz)
```

The values of 2-norm of the residual vector at each shift points are returned.

### 5.1.7 \*\_finalize

Release memories of the arrays stored in the library.

Syntax

Fortran

```
CALL komega_cg_r_finalize()  
CALL komega_cg_c_finalize()  
CALL komega_cocg_finalize()  
CALL komega_bicg_finalize()
```

C/C++

```

komega_cg_r_finalize();
komega_cg_c_finalize();
komega_cocg_finalize();
komega_bicg_finalize();

```

## 5.2 Sample codes for using shifted BiCG library

As a typical example, the usage of shifted BiCG library is shown below.

```

PROGRAM my_prog
!
USE komega_bicg, ONLY : komega_bicg_init, komega_bicg_restart, &
&
& komega_bicg_update, komega_bicg_getcoef, &
&
& komega_bicg_getvec, komega_bicg_finalize
USE solve_cc_routines, ONLY : input_size, input_restart, &
&
& projection, &
&
& hamiltonian_prod, generate_system, &
&
& output_restart, output_result
!
IMPLICIT NONE
!
INTEGER, SAVE :: &
& ndim, & ! Size of Hilvert space
& nz, & ! Number of frequencies
& nl, & ! Number of Left vector
& itermax, & ! Max. number of iteration
& iter_old ! Number of iteration of previous run
!
REAL(8), SAVE :: &
& threshold ! Convergence Threshold
!
COMPLEX(8), SAVE :: &
& z_seed ! Seed frequency
!
COMPLEX(8), ALLOCATABLE, SAVE :: &
& z(:) ! (nz): Frequency
!
COMPLEX(8), ALLOCATABLE, SAVE :: &
& ham(:, :), &
& rhs(:), &
& v12(:), v2(:), & ! (ndim): Working vector
& v14(:), v4(:), & ! (ndim): Working vector
& r_1(:), & ! (nl) : Projeccted residual vector
& x(:, :) ! (nl, nz) : Projected result
!
! Variables for Restart
!
COMPLEX(8), ALLOCATABLE, SAVE :: &
& alpha(:), beta(:) ! (iter_old)
!
COMPLEX(8), ALLOCATABLE, SAVE :: &
& r_1_save(:, :) ! (nl, iter_old) Projected residual vectors
!
! Variables for Restart
!
INTEGER :: &

```

```
& iter,      & ! Counter for Iteration
& status(3)
!
LOGICAL :: &
& restart_in, & ! If .TRUE., restart from the previous result
& restart_out  ! If .TRUE., save data for the next run
!
! Input Size of vectors, numerical conditions
!
CALL input_size(ndim,nl,nz)
CALL input_condition(itermax,threshold,restart_in,restart_out)
!
ALLOCATE(v12(ndim), v2(ndim), v14(ndim), v4(ndim), r_l(nl), &
&          x(nl,nz), z(nz), ham(ndim,ndim), rhs(ndim))
!
CALL generate_system(ndim, ham, rhs, z)
!
WRITE(*,*)
WRITE(*,*) "##### CG Initialization #####"
WRITE(*,*)
!
IF(restart_in) THEN
!
CALL input_restart(iter_old, zseed, alpha, beta, r_l_save)
!
IF(restart_out) THEN
CALL komega_bicg_restart( &
&   ndim, nl, nz, x, z, itermax, threshold, &
&   status, iter_old, v2, v12, v4, v14, alpha, &
&   beta, z_seed, r_l_save)
ELSE
CALL komega_bicg_restart( &
&   ndim, nl, nz, x, z, 0, threshold, &
&   status, iter_old, v2, v12, v4, v14, alpha, &
&   beta, z_seed, r_l_save)
END IF
!
! These vectors were saved in BiCG routine
!
DEALLOCATE(alpha, beta, r_l_save)
!
IF(status(1) /= 0) GOTO 10
!
ELSE
!
! Generate Right Hand Side Vector
!
v2(1:ndim) = rhs(1:ndim)
v4(1:ndim) = CONJG(v2(1:ndim))
!v4(1:ndim) = v2(1:ndim)
!
IF(restart_out) THEN
CALL komega_bicg_init(ndim, nl, nz, x, z, termx, threshold)
ELSE
CALL komega_bicg_init(ndim, nl, nz, x, z, 0, threshold)
END IF
!
END IF
```



```

!
! BiCG Loop
!
WRITE(*,*)
WRITE(*,*) "##### CG Iteration #####"
WRITE(*,*)
!
DO iter = 1, itermax
!
! Projection of Residual vector into the space
! spanned by left vectors
!
r_l(1:nl) = projection(v2(1:nl))
!
! Matrix-vector product
!
CALL hamiltonian_prod(Ham, v2, v12)
CALL hamiltonian_prod(Ham, v4, v14)
!
! Update result x with BiCG
!
CALL komega_bicg_update(v12, v2, v14, v4, x, r_l, status)
!
WRITE(*, '(a,i,a,3i,a,e15.5)') "lopp : ", iter, &
& " , status : ", status(1:3), &
& " , Res. : ", DBLE(v12(1))
IF(status(1) < 0) EXIT
!
END DO
!
IF(status(2) == 0) THEN
WRITE(*,*) " Converged in iteration ", ABS(status(1))
ELSE IF(status(2) == 1) THEN
WRITE(*,*) " Not Converged in iteration ", ABS(status(1))
ELSE IF(status(2) == 2) THEN
WRITE(*,*) " Alpha becomes infinity", ABS(status(1))
ELSE IF(status(2) == 3) THEN
WRITE(*,*) " Pi_seed becomes zero", ABS(status(1))
ELSE IF(status(2) == 4) THEN
WRITE(*,*) " Residual & Shadow residual are orthogonal", &
& ABS(status(1))
END IF
!
! Total number of iteration
!
iter_old = ABS(status(1))
!
! Get these vectors for restart in the Next run
!
IF(restart_out) THEN
!
ALLOCATE(alpha(iter_old), beta(iter_old), r_l_save(nl,iter_old))
!
CALL komega_bicg_getcoef(alpha, beta, z_seed, r_l_save)
CALL komega_bicg_getvec(v12,v14)
!
CALL output_restart(iter_old, z_seed, alpha, beta, &
& r_l_save, v12, v14)

```

```
        !
        DEALLOCATE(alpha, beta, r_l_save)
        !
    END IF
    !
10 CONTINUE
    !
    ! Deallocate all intrinsic vectors
    !
    CALL komega_bicg_finalize()
    !
    ! Output to a file
    !
    CALL output_result(nl, nz, z, x, r_l)
    !
    DEALLOCATE(v12, v2, v14, v4, r_l, x, z)
    !
    WRITE(*,*)
    WRITE(*,*) "##### Done #####"
    WRITE(*,*)
    !
END PROGRAM my_prog
```

---

## Re-distribution of this library

---

### 6.1 Contain Komega in your program

$K\omega$  library is distributed with the *Lesser General Public License* (LGPL). It is summarized as follows:

- $K\omega$  can be freely distributed, modified, copied and pasted, in a private program (in the research group, co-workers, etc.).
- For the released program (open-source, free, commercial software etc.):
  - When you contain the source-code of  $K\omega$  (either as is and modified) in the distributed source code of your program, please distribute your program with LGPL/GPL.
  - If you do not include the source-code of  $K\omega$  (just call it), you can freely distribute your program with any licenses.
  - If you distribute a binary file which is statically linked to  $K\omega$  library, please use LGPL/GPL. However, if you distribute a binary file which is dynamically linked to  $K\omega$  library (therefore  $K\omega$  itself is not contained), you can freely distribute your binary file with any licenses.

### 6.2 Build Komega without Autoconf

In this package,  $K\omega$  is built with Autotools (Autoconf, Automake, Libtool). If you do not want to use Autotools for your distributed program with  $K\omega$  source, you can use the following simple Makefile (please care about TAB).

```
F90 = gfortran
FFLAGS = -fopenmp -g -O2 #-D__MPI -D__NO_ZDOT -D__KOMECA_THREAD

.SUFFIXES :
.SUFFIXES : .o .F90

OBJS = \
komeca_cg_c.o \
komeca_cg_r.o \
komeca_cocg.o \
komeca_bicg.o \
komeca_math.o \
komeca_vals.o

all:libkomeca.a

libkomeca.a:$(OBJS)
```

```
    ar cr libkomega.a $(OBJS)

.F90.o:
    $(F90) -c $(FFLAGS)

clean:
    rm -f *.o *.a *.mod

komega_cg_c.o:komega_math.o
komega_cg_c.o:komega_vals.o
komega_cg_r.o:komega_math.o
komega_cg_r.o:komega_vals.o
komega_cocg.o:komega_math.o
komega_cocg.o:komega_vals.o
komega_bicg.o:komega_math.o
komega_bicg.o:komega_vals.o
komega_math.o:komega_vals.o
```

Preprocessor macros `__MPI`, `__NO_ZDOT`, and `__KOMEГА_THREAD` correspond to `--with-mpi=yes`, `--disable-zdot`, and `--enable-thread` of the options of `configure`, respectively.

## 6.3 Lesser General Public License

© 2016- The University of Tokyo. All rights reserved.

This software is developed under the support of  
“*Project for advancement of software usability in materials science*” by The  
Institute for Solid State Physics, The University of Tokyo.

This library is free software; you can redistribute it and/or  
modify it under the terms of the GNU Lesser General Public  
License as published by the Free Software Foundation; either  
version 2.1 of the License, or (at your option) any later version.  
This library is distributed in the hope that it will be useful,  
but WITHOUT ANY WARRANTY; without even the implied warranty of  
MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the GNU  
Lesser General Public License for more details.

You should have received a copy of the GNU Lesser General Public  
License along with this library; if not, write to the Free Software  
Foundation, Inc., 59 Temple Place, Suite 330, Boston, MA 02111-1307 USA

For more details, See ‘COPYING.LESSER’ in the root directory of this library.

---

## Contact

---

If you have any comments, questions, bug reports etc. about this library, please contact to the main developer (Mitsuaki Kawamura) by sending the e-mail (the address is shown below).

mkawamura\_at\_issp.u-tokyo.ac.jp

Please change \_at\_ into @, when you will send the e-mail.

---

Reference

---

- [1] A. Frommer, *Computing* **70**, 87 (2003).
- [2] S. Yamamoto, T. Sogabe, T. Hoshi, S.-L. Zhang, and T. Fujiwara, *J. Phys. Soc. Jpn.* **77**, 114713 (2008).